

ABAP. Новое в OpenSQL

Webinar

Василий Ковальский, SAP CIS.
25 июля, 2018

PUBLIC


Василий Ковальский,

инструктор ABAP с 1998 года

education.russia@sap.com




Серия вебинаров Обучение АВАР.



24 июля. Инлайновая декларация

<https://events.webinar.ru/5126393/1314773>



25 июля. Новый синтаксис OpenSQL

<https://events.webinar.ru/5126393/1314829>



26 июля. Tabstrip (панель вкладок) на селекционных экранах

<https://events.webinar.ru/5126393/1315175>



31 июля. Динамическая модификация селекционных экранов

<https://events.webinar.ru/5126393/1315179>



2 августа. Табличная буферизация

<https://events.webinar.ru/5126393/1315185>

О чем речь

Начиная с версии 7.4 появился новый синтаксис OpenSQL

Содержание

- 1. Новый синтаксис**
- 2. Вычисляемые поля**
- 3. Где еще об этом узнать?**

Новый синтаксис OpenSQL

1. Поля во фразах **SELECT**, **GROUP BY**, **ORDER BY** разделяются запятыми, хост-переменные начинаются с @
2. Инлайновая декларация переменной – приемника
3. **RIGHT [OUTER] JOIN** (с версии 7.4)
UNION (с версии 7.5)
4. Вычисляемые поля

Новый синтаксис OpenSQL

```
SELECT carrid,connid,forcurkey,  
       COUNT(*) AS count,SUM( forcuram ) AS forcuram  
FROM sbook  
INTO CORRESPONDING FIELDS OF @gs  
WHERE forcurkey IN @s_curr  
       AND carrid IN @s_carr  
GROUP BY carrid,connid,forcurkey  
ORDER BY carrid,connid,forcurkey  
.  
WRITE: / gs-carrid, gs-connid, gs-count, gs-forcuram, gs-forcurkey.  
ENDSELECT.
```

Инлайновые декларации во фразе INTO

```
SELECT carrid,connid,forcurkey,forcuram  
  FROM sbook UP TO 5 ROWS INTO @DATA(gs).  
  WRITE: / gs-carrid,gs-connid,gs-forcuram,gs-forcurkey.  
ENDSELECT.
```

SKIP.

```
SELECT carrid,connid,forcurkey,forcuram  
  FROM sbook UP TO 5 ROWS INTO TABLE @DATA(gt).  
LOOP AT gt INTO gs.  
  WRITE: / gs-carrid,gs-connid,gs-forcuram,gs-forcurkey.  
ENDLOOP.
```

UNION (с версии 7.5)

```
SELECT
    'F-module | ' AS tp
    ,funcname AS name
FROM tfdir
WHERE funcname LIKE @p_srch
UNION
SELECT
    'method | ' AS tp
    ,concat( classname, concat( '==-->',methodname ) ) AS name
FROM tmdir
WHERE methodname LIKE @p_srch
INTO TABLE @DATA(qqq)
.

LOOP AT qqq INTO DATA(www) .
    WRITE: / www.
ENDLOOP.
```


Вычисляемые поля. Строковые и арифметические выражения

```
SELECT @icon_ws_plane && carrid AS carrier
       ,connid
       ,fldate
       ,( seatsmax - seatsocc ) AS free
FROM sflight INTO @DATA(gs)
.
WRITE: / gs-carrier,gs-connid,gs-fldate,gs-free.
ENDSELECT.
```

Операции и функции в вычисляемых полях

С версии	Оператор или функция
7.40	+, -, *, /, CASE, скобки
7.40 SP05	&& ABS(), CEIL(), FLOOR(), DIV(), MOD(),
7.50	ROUND(), CONCAT(), CONCAT_WITH_SPACE(), LPAD(), LENGTH(), LTRIM(), RTRIM(), REPLACE(), RIGHT(), SUBSTRING()

&& до 255 символов

CONCAT() до 1333 символов

Вычисляемые поля. CASE-выражения

```
SELECT DISTINCT class,  
       CASE class  
       WHEN 'Y' THEN 'Economy'  
       WHEN 'C' THEN 'Business'  
       WHEN 'F' THEN 'First'  
       END AS txt  
FROM sbook INTO @DATA(gs).  
WRITE: / gs-class, gs-txt.  
ENDSELECT.
```

Вычисляемые поля. CAST-выражения. Приведение типа

```
SELECT SINGLE
*      carrid && connid                                " IMPOSSIBLE
*      carrid && cast( connid as char( 4 ) ) as txt " IMPOSSIBLE
concat( carrid , CAST ( connid AS CHAR( 4 ) ) ) AS txt
FROM   sflight
INTO @DATA(gs)
.
WRITE: / ':', gs COLOR COL_TOTAL.
```

CAST- выражения. Правила приведения «словарских» типов данных

Значение источника должно помещаться в приемник. **В остальном** действуют следующие правила:

from/to	INT1	INT2	INT4	INT8	DEC	CURR	QUAN	FLTP	CHAR	NUMC	CLNT	LANG	DATS	TIMS	UNIT	CUKY	SSTRING	RAW
INT1	X	X	X	X	X	X	X	X	Z	-	-	-	-	-	-	-	-	-
INT2	X	X	X	X	X	X	X	X	Z	-	-	-	-	-	-	-	-	-
INT4	X	X	X	X	X	X	X	X	Z	-	-	-	-	-	-	-	-	-
INT8	X	X	X	X	X	X	X	X	Z	-	-	-	-	-	-	-	-	-
DEC	-	-	-	-	X	X	X	X	Z	-	-	-	-	-	-	-	-	-
CURR	-	-	-	-	X	X	X	X	Z	-	-	-	-	-	-	-	-	-
QUAN	-	-	-	-	X	X	X	X	Z	-	-	-	-	-	-	-	-	-
FLTP	-	-	-	-	-	-	-	X	-	-	-	-	-	-	-	-	-	-
CHAR	-	-	-	-	-	-	-	-	X	y	y	y	y	y	x	y	-	-
NUMC	X	X	X	X	X	X	X	X	y	y	y	-	y	y	-	-	-	-
CLNT	-	-	-	-	-	-	-	-	X	y	x	-	-	-	-	-	-	-
LANG	-	-	-	-	-	-	-	-	X	-	-	X	-	-	-	-	-	-
DATS	-	-	-	-	-	-	-	-	X	y	-	-	X	-	-	-	-	-
TIMS	-	-	-	-	-	-	-	-	X	y	-	-	-	X	-	-	-	-
UNIT	-	-	-	-	-	-	-	-	X	-	-	-	-	-	X	-	-	-
CUKY	-	-	-	-	-	-	-	-	X	-	-	-	-	-	-	X	-	-
SSTRING	-	-	-	-	-	-	-	-	-	-	-	-	-	-	-	-	y	-
RAW	-	-	-	-	-	-	-	-	-	-	-	-	-	-	-	-	-	c

x – возможно без дополнительных ограничений.

y – длина источника должна быть не меньше чем приемника.

z – длина приемника должна быть достаточной для размещения значения источника.

c – источник и приемник должны быть одинаковой длины.

Где еще об этом узнать: Семинар D75AW

Семинар D75AW (3 дня)

Официальное описание,
регистрация на семинар (26 - 28 декабря 2018) :

<https://training.sap.com/course/d75aw--ru-ru>



Мое описание семинара

<https://sapland.ru/kb/articles/stats/novosti-abap-d75aw.html>


Где еще об этом узнать: Paul Hardy. - ABAP to the Future

Paul Hardy
ABAP to the Future

<https://sapland.ru/books/abap-to-the-future.html>



Где еще об этом узнать: Транзакция ABAPDOCU



Спасибо за внимание!

Василий Ковальский

Ведущий тренер по ABAP | SAP Education
education.russia@sap.com